Differential Geometry Routines

Douglas R.Lanman

January 9, 2006

dlanman@gmail.com

Define independent variables.

```
u = {ξ, η};
```

Define three-dimensional surface.

```
r = {ξ, η, ξ^2 - η^2};
```

Plot three-dimensional surface.

```
ParametricPlot3D[r, {ξ, -1, 1}, {η, -1, 1}];
```



Evaluate covariant base vectors.

```
g = {∂u[[1]] r, ∂u[[2]] r}
```

$$\{\{1, 0, 2\xi\}, \{0, 1, -2\eta\}\}$$

Determine covariant metric tensor components.

```
G = Table[g[[i]].g[[j]], {i, 1, 2}, {j, 1, 2}] // FullSimplify;
MatrixForm[G]
```

$$\begin{pmatrix} 1+4\xi^2 & -4\eta\xi \\ -4\eta\xi & 1+4\eta^2 \end{pmatrix}$$

Evaluate contravariant metric tensor components.

```
Gu = Inverse[G] // FullSimplify;
MatrixForm[Gu]
```

$$\begin{pmatrix} \frac{1+4\eta^2}{1+4\eta^2+4\xi^2} & \frac{4\eta\xi}{1+4\eta^2+4\xi^2} \\ \frac{4\eta\xi}{1+4\eta^2+4\xi^2} & \frac{1+4\xi^2}{1+4\eta^2+4\xi^2} \end{pmatrix}$$

Determine the contravariant metric tensor components.

```
gu = Table[Sum[Gu[[i, j]] g[[j]], {j, 1, 2}], {i, 1, 2}] // FullSimplify
```

$$\left\{\left\{\frac{1+4\eta^2}{1+4\eta^2+4\xi^2}, \frac{4\eta\xi}{1+4\eta^2+4\xi^2}, \frac{2\xi}{1+4\eta^2+4\xi^2}\right\}, \left\{\frac{4\eta\xi}{1+4\eta^2+4\xi^2}, \frac{1+4\xi^2}{1+4\eta^2+4\xi^2}, -\frac{2\eta}{1+4\eta^2+4\xi^2}\right\}\right\}$$

Verify contravariant and covariant basis vectors.

```
Table[g[[i]].gu[[j]], {i, 1, 2}, {j, 1, 2}] // FullSimplify //
 MatrixForm
```

$$\begin{pmatrix} 1 & 0 \\ 0 & 1 \end{pmatrix}$$

Evaluate surface normal.

$$n = \frac{\text{Cross}[g[[1]], g[[2]]]}{\sqrt{\text{Cross}[g[[1]], g[[2]]].\text{Cross}[g[[1]], g[[2]]]}} \text{ // FullSimplify}$$

$$\left\{-\frac{2\xi}{\sqrt{1+4\eta^2+4\xi^2}}, \frac{2\eta}{\sqrt{1+4\eta^2+4\xi^2}}, \frac{1}{\sqrt{1+4\eta^2+4\xi^2}}\right\}$$

Evaluate Christoffel symbols.

```
Γ[α_, β_, λ_] := (∂_u[[β]] g[[α]]).gu[[λ]];
Table[{α, β, λ, Γ[α, β, λ]}, {α, 1, 2}, {β, 1, 2}, {λ, 1, 2}]
```

$$\left\{\left\{\left\{\left\{1, 1, 1, \frac{4\xi}{1+4\eta^2+4\xi^2}\right\}, \left\{1, 1, 2, -\frac{4\eta}{1+4\eta^2+4\xi^2}\right\}\right\}, \{\{1, 2, 1, 0\}, \{1, 2, 2, 0\}\}\right\}, \left\{\{\{2, 1, 1, 0\}, \{2, 1, 2, 0\}\}, \left\{\left\{2, 2, 1, -\frac{4\xi}{1+4\eta^2+4\xi^2}\right\}, \left\{2, 2, 2, \frac{4\eta}{1+4\eta^2+4\xi^2}\right\}\right\}\right\}\right\}$$

Determine curvature tensor components.

```
B = Table[-(∂_u[[β]] n).g[[α]], {α, 1, 2}, {β, 1, 2}];
MatrixForm[B] // FullSimplify
```

$$\begin{pmatrix} \frac{2}{\sqrt{1+4\eta^2+4\xi^2}} & 0 \\ 0 & -\frac{2}{\sqrt{1+4\eta^2+4\xi^2}} \end{pmatrix}$$

Determine principal curvature (i.e. solve the generalized eigenvalue problem).

```
Curvature[a_, b_] := Solve[Evaluate[Det[B - σ G] == 0 /. {ξ → a, η → b}], σ]
```

Evaluate principal curvature at ξ,η = (0,0).

```
Curvature[0, 0]
```

$$\{\{\sigma \to -2\}, \{\sigma \to 2\}\}$$

Evaluate principal curvature at ξ,η = (1,0).

```
Curvature[1, 0]
```

$$\left\{\left\{\sigma \to -\frac{2}{\sqrt{5}}\right\}, \left\{\sigma \to \frac{2}{5\sqrt{5}}\right\}\right\}$$

Evaluate principal curvature at $\xi,\eta = (0,1)$.

```
Curvature[0, 1]
```

$$\left\{\left\{\sigma \to -\frac{2}{5\sqrt{5}}\right\}, \left\{\sigma \to \frac{2}{\sqrt{5}}\right\}\right\}$$

Validate result at $\xi,\eta = (0,0)$.

```
{D[(ξ^2 - η^2), {ξ, 2}], D[(ξ^2 - η^2), {η, 2}]}
```

```
{2, -2}
```